Geodesic Motion in the Schwarzschild Metric

The Geodesic Equations for Massless Particles

The geodesic motion of freely falling particles can be obtained from the solution of the geodesic equations. We choose the coordinates so that the motion is confined in the $\theta = \frac{\pi}{2}$ plane. Therefore $\dot{\theta} = 0,\ \sin(\theta) = 1$. Then we have to solve the system

$$\ddot{r} = \frac{\dot{r}^2 - e^2}{r^2\left(1 - \frac{2}{r}\right)} + \frac{1}{r^3}\left(1 - \frac{2}{r}\right)$$

$$\dot{t} = \frac{e}{1 - \frac{2}{r}}$$

$$\dot{\phi} = \frac{1}{r^2}$$

where $\dot{t} = \frac{d}{d\lambda}\, t(\lambda)$ etc, and the conserved quantities

$$e = \left(1 - \frac{2}{r}\right)\dot{t}$$

$$1 = r^2\,\dot{\phi}$$

are given.

We have rescaled the affine parameter of the null geodesic, so that the angular momentum $l = 1$.

The radial motion "conservation of energy" is given by

$$e^2 = \frac{1}{2}\,\dot{r}^2 + W_{\text{eff}}(r)$$

$$W_{\text{eff}}(r) = \frac{1}{r^2}\left(1 - \frac{2}{r}\right)$$

The turning points of the radial motion are calculated from the equation

$$W_{\text{eff}}(r) = e^2$$

The maximum of $W_{\text{eff}}(r)$ is at $r_{\max} = 3$ and $W_{\text{eff}}(r_{\max}) = \frac{1}{27}$. Therefore, we have circular orbits when $e^2 = \frac{1}{27},\ b = \sqrt{27}$

The impact parameter is

$$b^2 = \frac{1}{e^2}$$

In the above equations, dimensionless quantities are used. To go back to geometrized units, use the dictionary:

First restore l:

$\lambda \to l\,\lambda \quad \frac{d}{d\lambda} \to \frac{1}{l}\,\frac{d}{d\lambda}$

$\dot{t} \to l^{-1}\,\dot{t} \;\Rightarrow e \to l^{-1}\,e$

Then restore M:

$\tau \to \frac{\tau}{M} \quad t \to \frac{t}{M} \quad r \to \frac{r}{M} \quad \phi \to \phi$

$l \to \frac{l}{M} \quad W_{\text{eff}} \to M^2\,W_{\text{eff}} \quad e \to e \quad b \to b$

Effective potential for radial motion of massless particle

In[ ]:=

```
l = 4.3; rmax = 8.;
Weff[r_] := 1/r^2 (1 - 2/r);
Plot[Evaluate[{
    Weff[r],
    Weff[rmax]
   }], {r, 2, 40},
  PlotRange   → {-0.005, 0.05},
  PlotLegends → {"l=" <> ToString[l], "e^2= " <> ToString[Weff[rmax]]},
  PlotLabel   → "Turning Points: " <>
    ToString[Select[r /. NSolve[Weff[r] == Weff[rmax] , r], # ∈ Reals && # > 0 &]] <>
    "  Impact parameter: b= 1/e = " <> ToString[1/Sqrt[Weff[rmax]]],
  ImageSize   → Large]
```

Out[ ]=



Geodesic Equations

In[ ]:=

```
Weff[r_] := 1/r^2 (1 - 2/r);

solveGeodesicEqs[e_, ϕ0_, r0_, vr0_, λfinal_] := NDSolve[{
    t'[τ]  == e/(1 - 2/r[τ]),
    ϕ'[τ]  == 1/r[τ]^2 ,
    r''[τ] == -e^2/(r[τ]^2 (1 - 2/r[τ])) + (r'[τ])^2/(r[τ]^2 (1 - 2/r[τ])) + (1 - 2/r[τ]) 1/r[τ]^3 ,
    t[0] == 0, ϕ[0] == ϕ0, r[0] == r0,
    r'[0] == vr0  (* initial conditions               *)
  }, {t, ϕ, r}, {τ, 0, λfinal}
 ];
```

Plotting function

Skip this section if you are not interested in the plotting commands.
This is the function that plots the results.

```
In[ ]:= plotResults[xyRange_] :=
 (* xyRange is the range of the x-y plot *)
 Module[{xyrange = xyRange},
  GraphicsGrid[
   {                                    (* Arrange the plots in a grid          *)
    { (* 1st row *)
     Show[{
       ParametricPlot[Evaluate[
         {r[τ] Cos[ϕ[τ]], r[τ] Sin[ϕ[τ]]} (* x=r cos(ϕ) y=r sin(ϕ)                *)
          /. sol], {τ, 0.1, λmax}, AxesLabel → {"x", "y"}, PlotRange → xyrange],
       Graphics[{Red, Disk[{0, 0}, 2]}]
       (* a red disk in the black hole area     *)
      }],
     Plot[{
       Weff[r], Energy
       (* The effective potential with the energy level E *)
      }, {r, 0, 40},
      PlotRange → {-0.001, 0.05},
      PlotLabel → "Turning Points: "              (* compute
         the turning points                         *)
        <> ToString[Select[r /. NSolve[Weff[r] == Energy , r], # ∈ Reals && # > 0 &]]
        <> "      e^2= " <> ToString[Energy]
        (*  display the value of E on the plot                *)
        <> "      b = " <> ToString[1/e],
      AxesLabel → {"r", "W_eff[r]"}]
    }, (* 1st row *)
    {  (* 2nd row *)
     Plot[Evaluate[{
         r[τ]
         (* radial  coordinate as a function of λ             *)
        } /. sol], {τ, 0, λmax}, AxesLabel → {"λ", "r[λ]"}, PlotRange → All],
     Plot[Evaluate[{
         Mod[ϕ[τ], 2 π]
```

```
          (* angular coordinate as a function of λ           *)
         }/. sol], {τ, 0, λmax}, AxesLabel → {"λ", "ϕ[λ]"}, PlotRange → All]
    }, (* 2nd row *)
    {  (* 3rd row *)
      Plot[Evaluate[{
          t[τ]
          (* time  coordinate as a function of λ              *)
         }/. sol], {τ, 0, λmax}, AxesLabel → {"λ", "t[λ]"}],
      Plot[Evaluate[
        (* check numerical errors in ℰ                         *)
        (r'[τ]^2 + Weff[r[τ]] - Energy) /. sol
       ], {τ, 0, λmax}, AxesLabel → {"λ", "Δℰ"}, PlotLabel → "Error in ℰ ≡ e^2"]
    } (* 3rd row *)

  }, ImageSize → Full]
]
```

Solutions to the geodesic equations for massless particles

Set:

r1: one of the turning points

e : $b^2 = 1/e^2$

r0: initial radial position

λmax: maximum time for numerical integration for the affine parameter λ

radialdirection: the sign of the initial radial velocity $\dot{r}(0)$

r1 will determine the conserved quantity e, which in turn will determine $|\,\dot{r}(0)\,|$

v0 = (radialdirection) × $|\,\dot{r}(0)\,|$

Circular trajectories

The unstable circular orbit for $r(0) = r_{\max} = 3,\ e^2 = W_{\text{eff}}(r_{\max}) = \frac{1}{\sqrt{27}}$

The numerical integration has an instability for $\tau \approx 159$

In[ ]:=

```
r1   = 3. ; λmax = 150 ;
r0   = 3. ; radialdirection = -1; (* Set it to +/- 1. The sign of ṙ(0)  *)

Energy = Weff[r1]; e = √Energy ; b = 1/e;
v0 = radialdirection √(Energy - Weff[r0]) ;
ϕ0 = 0;

sol = solveGeodesicEqs[e, ϕ0, r0, v0, λmax]; gg = plotResults[All]
```

Out[ ]=



Instability of the circular orbits:

Slightly perturbed away from the $r_{max} = 3$, $W_{max} = \frac{1}{27}$ circular orbit

In[∘]:=

```
r1   = 3.0001 ; λmax = 150 ;
r0   = 3.0001 ; radialdirection = -1; (* Set it to +/- 1. The sign of ṙ(0)  *)

Energy = Weff[r1]; e = √Energy ; b = 1/e;
v0 = radialdirection √(Energy - Weff[r0]) ;
ϕ0 = 0;

sol = solveGeodesicEqs[e, ϕ0, r0, v0, λmax]; gg = plotResults[All]
```

Out[∘]=





Scattering - Deflection of light

Set:

b: the impact parameter

x0: initial x coordinate

V0: v_0, the x-component of the velocity



Then:

$r_0 \quad = \sqrt{{x_0}^2 + b^2}$

$v_0 \quad = 1 = \frac{\mathrm{dx}}{\mathrm{dt}} \qquad$ (far from $r = 0$)

$v^r \quad =$

$-v_0 \cos(\phi_0) = \frac{\mathrm{dr}}{\mathrm{dt}} = \frac{\mathrm{dr}}{\mathrm{d}\lambda} \Big/ \frac{\mathrm{dt}}{\mathrm{d}\lambda} = \dot{r}(0) \Big/ \left(e\left(1 - \frac{2}{r}\right)^{-1}\right) \Rightarrow \dot{r}(0) = -e\left(1 - \frac{2}{r}\right)^{-1} \cos(\phi_0) \Rightarrow \dot{r}(0) = -\frac{1}{b}\left(1 - \frac{2}{r}\right)^{-1} \cos(\phi_0)$

$e^2 \quad = {\dot{r}_0}^2 + W_{\mathrm{eff}}(r_0)$

To have finite deflection angle, we must have $b > \sqrt{27} \approx 5.196$

```
b  = 8; x0 = 500.; λmax = 12 000 ;

r0 =  √(x0^2 + b^2) ;
ϕ0 = ArcTan[b/x0];
v0  = -1/b (1 - 2/r0)^-1 Cos[ϕ0];

Energy = v0^2 + Weff[r0];
e = √Energy ; b = 1 / e; (* b must be redefined,
it is only approximately equal to the impact parameter *)

sol = solveGeodesicEqs[e, ϕ0, r0, v0, λmax];
gg = plotResults[{{-40, 100}, {-40, 10}}];

Print[
  "Scattering Angle: (degrees)\n",
  "θ= tan^-1 v^y (∞)/v^x (∞)  = ",
  Δθ = 360 + 180/π (ArcTan[r '[λmax] Cos[ϕ[λmax]] - ϕ '[λmax] r[λmax] Sin[ϕ[λmax]] /. sol,
         r '[λmax] Sin[ϕ[λmax]] + ϕ '[λmax] r[λmax] Cos[ϕ[λmax]] /. sol ]) , "\n",
  "δθ_deflection= ", Δθ - 180
 ];
Show[gg]
```

```
Scattering Angle: (degrees)
θ= tan^-1 v^y (∞)/v^x (∞)  = {229.575}
δθ_deflection= {49.5747}
```

Out[ ]=


ng Points: {6.66648, 2.15837} e2= 0.0157507
y
x
Weff[r]
r
r[λ]
λ
ϕ[λ]
λ
t[λ]
λ
Error in ℰ ≡ e2
Δℰ
λ

In[ ]:=

```
b  = 5.3; x0 = 500.; λmax = 12 000 ;

r0 = Sqrt[x0^2 + b^2];
ϕ0 = ArcTan[b/x0];
v0 = -(1/b) (1 - 2/r0)^-1 Cos[ϕ0];
Energy = v0^2 + Weff[r0]; e = Sqrt[Energy]; b = 1/e;

sol = solveGeodesicEqs[e, ϕ0, r0, v0, λmax];
gg = plotResults[{{-10, 100}, {-10, 70}}]
```

Out[ ]=



In[ ]:=
```
b  = 5.1; x0 = 500.; λmax = 2567 ;

r0 = Sqrt[x0^2 + b^2] ;

ϕ0 = ArcTan[b/x0];

v0  = -1/b (1 - 2/r0)^-1 Cos[ϕ0];


Weff[r_] := 1/r^2 (1 - 2/r);

Energy = v0^2 + Weff[r0];

e = Sqrt[Energy] ; b = 1 / e; (* b must be redefined,
it is only approximately equal to the impact parameter *)

sol = solveGeodesicEqs[e, ϕ0, r0, v0, λmax];
gg  = plotResults[{{-5, 30}, {-10, 10}}]
```

Out[ ]=


Turning Points: {}
e²= 0.0387561
b = 5.079
y
x
W_eff[r]
r
r[λ]
λ
ϕ[λ]
t[λ]
Error in ℰ ≡ e²
Δℰ

In[ ]:=
```
b = 5.2170155293; x0 = 500.; λmax = 3000 ;

r0 = Sqrt[x0^2 + b^2] ;
ϕ0 = ArcTan[b/x0];
v0 = -(1/b) (1 - 2/r0)^-1 Cos[ϕ0];

Energy = v0^2 + Weff[r0];
e = Sqrt[Energy] ; b = 1/e; (* b must be redefined,
it is only approximately equal to the impact parameter *)

sol = solveGeodesicEqs[e, ϕ0, r0, v0, λmax];
gg = plotResults[{{-5, 30}, {-10, 10}}];

Print[
  "Scattering Angle: (degrees)\n",
  "θ= tan^-1 v^y(∞)/v^x(∞) = ",
  Δθ = 360 + 180/π (ArcTan[r '[λmax] Cos[ϕ[λmax]] - ϕ '[λmax] r[λmax] Sin[ϕ[λmax]] /. sol,
        r '[λmax] Sin[ϕ[λmax]] + ϕ '[λmax] r[λmax] Cos[ϕ[λmax]] /. sol ]) , "\n",
  "δθ_deflection= ", Δθ - 180
 ];
Show[gg]
```

Scattering Angle: (degrees)

$\theta = \tan^{-1}\frac{v^y(\infty)}{v^x(\infty)}$ = {324.304}

$\delta\theta_{\text{deflection}}$ = {144.304}

Out[ ]=



Acknowledgements

This notebook has been programmed by Konstantinos Anagnostopoulos, Physics Department, National Technical University of Athens, Greece, while he was an instructor of the 4th year undergraduate course "General Relativity and Cosmology". It was created for fun, but it may turn out to be useful to everyone studying the General Theory of Relativity for the first time.

Email: konstant@mail.ntua.gr

Web: http://physics.ntua.gr/konstant